Problem:
Score : 100 points

Problem Statement

There are N items, numbered 1, 2, \ldots, N.
For each i (1 \leq i \leq N), Item i has a weight of w_i and a value of v_i.

Taro has decided to choose some of the N items and carry them home in a knapsack.
The capacity of the knapsack is W, which means that the sum of the weights of items taken must be at most W.

Find the maximum possible sum of the values of items that Taro takes home.

Constraints

All values in input are integers.

1 \leq N \leq 100

1 \leq W \leq 10^9

1 \leq w_i \leq W

1 \leq v_i \leq 10^3

Input

Input is given from Standard Input in the following format:

N W
w_1 v_1
w_2 v_2
:
w_N v_N

Output

Print the maximum possible sum of the values of items that Taro takes home.

Sample Input 1

3 8
3 30
4 50
5 60

Sample Output 1

90

Items 1 and 3 should be taken.
Then, the sum of the weights is 3 + 5 = 8, and the sum of the values is 30 + 60 = 90.

Sample Input 2

1 1000000000
1000000000 10

Sample Output 2

10

Sample Input 3

6 15
6 5
5 6
6 4
6 6
3 5
7 2

Sample Output 3

17

Items 2, 4 and 5 should be taken.
Then, the sum of the weights is 5 + 6 + 3 = 14, and the sum of the values is 6 + 6 + 5 = 17.

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import qualified Data.ByteString.Char8 as B
import Data.List
import qualified Data.Vector.Unboxed as V
readIntPair :: B.ByteString -> (Int,Int)
readIntPair s = p
    where
    Just p = do
        (n1, s') <- B.readInt s
        (n2, _ ) <- B.readInt (B.tail s')
        return (n1,n2)

(&) :: a -> (a -> b) -> b
(&) a b = b a

infixl 1 &

main :: IO ()
main = do
    (n,w_limit) <- fmap readIntPair B.getLine
    items <- fmap (map readIntPair . B.lines) B.getContents 
    let v_limit = 1000 * n + 1
        v_end = items 
                    & foldl' step s0 
                    & V.findIndices (<=w_limit) 
                    & V.maximum
        inf = w_limit + 1
        step :: V.Vector Int -> (Int, Int) -> V.Vector Int
        step tbl (wi, vi) = tbl'
            where
            tbl' = V.generate v_limit $ \v -> 
                    min (tbl V.! v) 
                        (if v >= vi then (tbl V.! (v - vi)) + wi 
                                    else w_limit+1)
        s0 = V.generate v_limit $ \i -> if i == 0 then 0 else inf
    print (v_end :: Int)